

Compiler Design
COMP442 - Winter 2021

Assignment 2

Report

Kresten Ordekian 40030197

Section 1. Transformed grammar into LL(1)

<AParams> ::= <Expr> <AParamsTail>

<AParams> ::= EPSILON

<AParamsTail> ::= ',' <Expr> <AParamsTail>

<AParamsTail> ::= EPSILON

<AddOp> ::= '+'

<AddOp> ::= '-'

<AddOp> ::= 'or'

<ArithExpr> ::= <Term> <ArithExprTail>

<ArithExprTail> ::= <AddOp> <Term> <ArithExprTail>

<ArithExprTail> ::= EPSILON

<ArraySizeRept> ::= '[' <IntNum> ']' <ArraySizeRept>

<ArraySizeRept> ::= EPSILON

<AssignOp> ::= 'assign'

<AssignStatTail> ::= <AssignOp> <Expr>

<ClassDecl> ::= 'class' 'id' <Inherit> '{' <ClassDeclBody> '}' ';' <ClassDecl>

<ClassDecl> ::= EPSILON

<ClassDeclBody> ::= <Visibility> <MemberDecl> <ClassDeclBody>

<ClassDeclBody> ::= EPSILON

<ClassMethod> ::= 'sr' 'id'

<ClassMethod> ::= EPSILON

<Expr> ::= <ArithExpr> <ExprTail>

<ExprTail> ::= <RelOp> <ArithExpr>

<ExprTail> ::= EPSILON

<FParams> ::= <Type> 'id' <ArraySizeRept> <FParamsTail>

<FParams> ::= EPSILON

<FParamsTail> ::= ',' <Type> 'id' <ArraySizeRept> <FParamsTail>

<FParamsTail> ::= EPSILON

<Factor> ::= <FuncOrVar>
 <Factor> ::= 'intnum'
 <Factor> ::= 'floatnum'
 <Factor> ::= 'stringlit'
 <Factor> ::= '(' <Expr> ')'
 <Factor> ::= 'not' <Factor>
 <Factor> ::= <Sign> <Factor>
 <Factor> ::= 'qm' '[' <Expr> ':' <Expr> ':' <Expr> ']'

<FuncBody> ::= '{' <MethodBodyVar> <StatementList> '}'

<FuncDecl> ::= 'func' 'id' '(' <FParams> ')' ':' <FuncDeclTail> ';'

<FuncDeclTail> ::= <Type>
 <FuncDeclTail> ::= 'void'

<FuncDef> ::= <Function> <FuncDef>
 <FuncDef> ::= EPSILON

<FuncHead> ::= 'func' 'id' <ClassMethod> '(' <FParams> ')' ':' <FuncDeclTail>

<FuncOrAssignStat> ::= 'id' <FuncOrAssignStatIdnest>

<FuncOrAssignStatIdnest> ::= <IndiceRep> <FuncOrAssignStatIdnestVarTail>
 <FuncOrAssignStatIdnest> ::= '(' <AParams> ')' <FuncOrAssignStatIdnestFuncTail>

<FuncOrAssignStatIdnestFuncTail> ::= '.' 'id' <FuncStatTail>
 <FuncOrAssignStatIdnestFuncTail> ::= EPSILON

<FuncStatTail> ::= <IndiceRep> '.' 'id' <FuncStatTail>
 <FuncStatTail> ::= '(' <AParams> ')' <FuncStatTailIdnest>

<FuncStatTailIdnest> ::= '.' 'id' <FuncStatTail>
 <FuncStatTailIdnest> ::= EPSILON

<FuncOrAssignStatIdnestVarTail> ::= '.' 'id' <FuncOrAssignStatIdnest>
 <FuncOrAssignStatIdnestVarTail> ::= <AssignStatTail>

<FuncOrVar> ::= 'id' <FuncOrVarIdnest>

<FuncOrVarIdnest> ::= <IndiceRep> <FuncOrVarIdnestTail>
 <FuncOrVarIdnest> ::= '(' <AParams> ')' <FuncOrVarIdnestTail>

<FuncOrVarIdnestTail> ::= '.' 'id' <FuncOrVarIdnest>

<FuncOrVarIdnestTail> ::= EPSILON

<Function> ::= <FuncHead> <FuncBody>

<IndiceRep> ::= '[' <Expr> ']' <IndiceRep>

<IndiceRep> ::= EPSILON

<Inherit> ::= 'inherits' 'id' <NestedId>

<Inherit> ::= EPSILON

<IntNum> ::= 'intnum'

<IntNum> ::= EPSILON

<MemberDecl> ::= <FuncDecl>

<MemberDecl> ::= <VarDecl>

<MethodBodyVar> ::= 'var' '{' <VarDeclRep> '}'

<MethodBodyVar> ::= EPSILON

<MultOp> ::= '*'

<MultOp> ::= '/'

<MultOp> ::= 'and'

<NestedId> ::= ',' 'id' <NestedId>

<NestedId> ::= EPSILON

<Prog> ::= <ClassDecl> <FuncDef> 'main' <FuncBody>

<RelOp> ::= 'eq'

<RelOp> ::= 'neq'

<RelOp> ::= 'lt'

<RelOp> ::= 'gt'

<RelOp> ::= 'leq'

<RelOp> ::= 'geq'

<START> ::= <Prog>

<Sign> ::= '+'

<Sign> ::= '-'

<StatBlock> ::= '{' <StatementList> '}'

<StatBlock> ::= <Statement>

<StatBlock> ::= EPSILON

<Statement> ::= <FuncOrAssignStat> ';'
<Statement> ::= 'if' '(' <Expr> ')' 'then' <StatBlock> 'else' <StatBlock> ';'
<Statement> ::= 'while' '(' <Expr> ')' <StatBlock> ';'
<Statement> ::= 'read' '(' <Variable> ')' ';'
<Statement> ::= 'write' '(' <Expr> ')' ';'
<Statement> ::= 'return' '(' <Expr> ')' ';'
<Statement> ::= 'break' ';'
<Statement> ::= 'continue' ';'

<StatementList> ::= <Statement> <StatementList>
<StatementList> ::= EPSILON

<Term> ::= <Factor> <TermTail>

<TermTail> ::= <MultOp> <Factor> <TermTail>
<TermTail> ::= EPSILON

<Type> ::= 'integer'
<Type> ::= 'float'
<Type> ::= 'string'
<Type> ::= 'id'

<VarDecl> ::= <Type> 'id' <ArraySizeRept> ';'

<VarDeclRep> ::= <VarDecl> <VarDeclRep>
<VarDeclRep> ::= EPSILON

<Variable> ::= 'id' <VariableIdnest>

<VariableIdnest> ::= <IndiceRep> <VariableIdnestTail>

<VariableIdnestTail> ::= '.' 'id' <VariableIdnest>
<VariableIdnestTail> ::= EPSILON

<Visibility> ::= 'public'
<Visibility> ::= 'private'
<Visibility> ::= EPSILON

Section 2. FIRST and FOLLOW sets

Symbol	First set	Follow set
<i>ADDOP</i>	plus, minus, or	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm, \$
<i>APARAMS</i>	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm	rpar
<i>APARAMSTAIL</i>	comma	rpar
<i>ARITHEXPR</i>	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm	comma, rsqbr, semi, rpar, colon, eq, neq, lt, gt, leq, geq
<i>ARITHEXPRTAIL</i>	plus, minus, or	comma, rsqbr, semi, rpar, colon, eq, neq, lt, gt, leq, geq
<i>ARRAYSIZEREPT</i>	lsqbr	comma, semi, rpar
<i>ASSIGNOP</i>	assign	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm
<i>ASSIGNSTATTAIL</i>	assign	semi
<i>CLASSDECL</i>	class	func, main
<i>CLASSDECLBODY</i>	id, func, integer, float, string, public, private	rcurbr
<i>CLASSMETHOD</i>	sr	lpar
<i>EXPR</i>	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm	comma, rsqbr, semi, rpar, colon
<i>EXPRTAIL</i>	eq, neq, lt, gt, leq, geq	comma, rsqbr, semi, rpar, colon
<i>FACTOR</i>	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm	plus, minus, or, comma, rsqbr, semi, rpar, colon, mult, div, and, eq, neq, lt, gt, leq, geq
<i>FPARAMS</i>	id, integer, float, string	rpar
<i>FPARAMSTAIL</i>	comma	rpar
<i>FUNCBODY</i>	lcurbr	func, main
<i>FUNCDECL</i>	func	id, rcurbr, func, integer, float, string, public, private

<i>FUNCDECLTAIL</i>	id, void, integer, float, string	lcurbr, semi
<i>FUNCDEF</i>	func	main
<i>FUNCHEAD</i>	func	lcurbr
<i>FUNCORASSIGNSTAT</i>	id	semi
<i>FUNCORASSIGNSTATID NEST</i>	lsqbr, assign, lpar, dot	semi
<i>FUNCORASSIGNSTATID NESTFUNCTAIL</i>	dot	semi
<i>FUNCORASSIGNSTATID NESTVARTAIL</i>	assign, dot	semi
<i>FUNCORVAR</i>	id	plus, minus, or, comma, rsqbr, semi, rpar, colon, mult, div, and, eq, neq, lt, gt, leq, geq
<i>FUNCORVARIDNEST</i>	lsqbr, lpar, dot	plus, minus, or, comma, rsqbr, semi, rpar, colon, mult, div, and, eq, neq, lt, gt, leq, geq
<i>FUNCORVARIDNESTTA IL</i>	dot	plus, minus, or, comma, rsqbr, semi, rpar, colon, mult, div, and, eq, neq, lt, gt, leq, geq
<i>FUNCSTATTAIL</i>	lsqbr, lpar, dot	semi
<i>FUNCSTATTAILIDNEST</i>	dot	semi
<i>FUNCTION</i>	func	func, main
<i>INDICEREP</i>	lsqbr	plus, minus, or, comma, rsqbr, assign, semi, rpar, colon, dot, mult, div, and, eq, neq, lt, gt, leq, geq
<i>INHERIT</i>	inherits	lcurbr
<i>INTNUM</i>	intnum	rsqbr
<i>MEMBERDECL</i>	id, func, integer, float, string	id, rcurbr, func, integer, float, string, public, private
<i>METHODBODYVAR</i>	var	id, rcurbr, if, while, read, write, return, break, continue
<i>MULTOP</i>	mult, div, and	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm

<i>NESTEDID</i>	comma	lcurbr
<i>PROG</i>	class, func, main	
<i>RELOP</i>	eq, neq, lt, gt, leq, geq	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm
<i>SIGN</i>	plus, minus	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm
<i>START</i>	class, func, main	
<i>STATBLOCK</i>	id, lcurbr, if, while, read, write, return, break, continue	semi, else
<i>STATEMENT</i>	id, if, while, read, write, return, break, continue	id, rcurbr, semi, if, else, while, read, write, return, break, continue
<i>STATEMENTLIST</i>	id, if, while, read, write, return, break, continue	rcurbr
<i>TERM</i>	plus, minus, id, intnum, floatnum, stringlit, lpar, not, qm	plus, minus, or, comma, rsqbr, semi, rpar, colon, eq, neq, lt, gt, leq, geq
<i>TERMTAIL</i>	mult, div, and	plus, minus, or, comma, rsqbr, semi, rpar, colon, eq, neq, lt, gt, leq, geq
<i>TYPE</i>	id, integer, float, string	id, lcurbr, semi
<i>VARDECL</i>	id, integer, float, string	id, rcurbr, func, integer, float, string, public, private
<i>VARDECLREP</i>	id, integer, float, string	rcurbr
<i>VARIABLE</i>	id	rpar
<i>VARIABLEIDNEST</i>	lsqbr, dot	rpar
<i>VARIABLEIDNESTTAIL</i>	dot	rpar
<i>VISIBILITY</i>	public, private	id, func, integer, float, string

Section 3. Design

I chose to implement a recursive descent predictive parser. I chose it because I wanted to be able to debug and see everything step by step.

I wrote a function for each of the non terminal symbols. I wrote a match function to check if the terminal symbols and tokens match.

I wrote skipErrors method to find and display a message of the line of the error to the user and recover from the error

I have a stack that keeps track of the function calls and I get the derivation output using the reverse of the stack and terminal symbols.

I have a separate ast.py file that has all the ast data structure methods.

Section 4. Use of tools

I used graphviz to visualize the tree.